



Eidgenössische Technische Hochschule Zürich  
Swiss Federal Institute of Technology Zurich

*Distributed  
Computing*



# **A Foundational Formalization of Grounded Arithmetic in Isabelle/Pure**

BSc Thesis

Sascha Kehrli  
skehrli@ethz.ch

Distributed Computing Group  
Computer Engineering and Networks Laboratory  
ETH Zürich

**Supervisors:**  
Prof. Dr. Bryan Ford  
Prof. Dr. Roger Wattenhofer

24.09.2025

# Abstract

# Contents

<b>1</b>	<b>Introduction</b>	<b>1</b>
1.1	Motivation: Recursive Definitions in Classical and Constructive Logic	1
1.2	Enter GD	2
<b>2</b>	<b>Background</b>	<b>4</b>
2.1	Isabelle/Pure	4
2.1.1	Syntax of Pure	4
2.1.2	Equality, Implication, and Quantification as Type Constructors	5
2.1.3	Deduction Rules	5
2.1.4	Formalizing Object Logics in Pure	6
2.2	Grounded Arithmetic (GA)	8
2.2.1	BGA Formalization	9
2.2.2	GA with Axiomatized Quantifiers	13
<b>3</b>	<b>Formalizing GA in Pure</b>	<b>14</b>
3.1	Propositional Axioms	14
3.2	Natural Number Axioms	15
3.3	Grounded Contradiction	18
3.4	Syntax Translation for Natural Numbers	18
3.5	Quantifier Axiomatization	19
3.6	Conditional Evaluation Axiomatization	20
3.7	Definitional Mechanism Axiomatization	21
3.8	Defining Arithmetic Functions in GA	22
3.9	Termination Proofs	23
<b>4</b>	<b>Tooling for Isabelle/GA</b>	<b>32</b>
4.1	(Un)folding (Recursive) Definitions	32
4.2	Configuring the Simplifier	33
<b>5</b>	<b>Encoding Inductive Datatypes in GD</b>	<b>36</b>
5.1	Next	36
<b>A</b>	<b>References</b>	<b>38</b>

# Introduction

1

## 1.1 Motivation: Recursive Definitions in Classical and Constructive Logic

We start with the simple observation that in logics of both classical and constructive tradition, there is a seemingly inherent lack of definitional freedom. That is, definitions must describe provably terminating expressions. The reason for these restrictions is that, without them in place, these logics would be inconsistent.

To see this for the case of classical logic, consider the definition

$$L \equiv \neg L.$$

Let us imagine that this is a valid definition in a classical logic (that is, a logic that at least has the law of excluded middle (LEM) and double negation elimination). If the logic allows us to deduce either of  $L$  or  $\neg L$ , the other can be deduced as well by unfolding the definition and making use of double negation elimination, making the logic inconsistent.

Thanks to the LEM, we can prove that  $L$  holds by contradiction.

Assuming  $\neg L$ , we can derive  $\neg\neg L$  by unfolding the definition once and then  $L$  via double negation elimination. Since we derived both  $L$  and  $\neg L$  from hypothetically assuming  $\neg L$ , a contradiction, this allows us to definitely conclude  $L$ .

What went wrong? The law of excluded middle forces a truth value on any term in classical logic, thus circular or non-sensical definitions such as  $L \equiv \neg L$ , for which no truth value can or should be assigned, cannot be admitted.

Constructive logics discard the law of excluded middle and are thus safe from a proof by contradiction like the one shown above. However, in intuitionistic tradition, lambda calculus terms are interpreted as proof terms, witnessing the truth of the proposition encoded by their type. Lambda functions of type  $A \Rightarrow B$  are then interpreted as producing a proof of  $B$  given a proof of  $A$ , which however means that they must always terminate.

To see this, consider the following attempt at a definition of an (ill-founded) term of type  $\forall\alpha.\alpha$ , i.e., a proof of every proposition:

$$\text{prove\_anything} := \Lambda\alpha. \text{prove\_anything } \alpha$$

Here, the construct  $\Lambda$  is the type-level analogue of lambda abstraction: it abstracts over a type variable and substitutes it in the body. That is, if  $e$  has type  $T$ , then  $\Lambda\alpha.e$  has type  $\forall\alpha.T$ .

If such a term were permitted in the logic, it would type-check as having type  $\forall\alpha.\alpha$ . Instantiating it at any type  $P$  yields a term of type  $P$ , i.e., a proof of  $P$  for arbitrary  $P$ , making every proposition in the logic trivially provable.

What went wrong this time? Functions in constructive logics represent logical implication. If a function has type  $A \Rightarrow B$ , the function must provide proof of  $B$ , that is, return a term  $b : B$ , given any term  $a : A$ . The function *witnesses* the implication of  $A$  to  $B$ . If the function does not terminate on an input however, this proof is not actually constructed and assuming the hypothetical resulting proof term leads to inconsistency.

## 1.2 Enter GD

Grounded deduction (GD) is a logical framework developed recently at EPFL and whose development was motivated by precisely the observation made above. The project aims to axiomatize a consistent (free from contradiction) formal system, in which arbitrary recursion in definitions is permitted and which is still as expressive as possible.

In Section 2.2, *Grounded Arithmetic (GA)*, a first-order theory of arithmetic based on *grounded* principles, is fully formalized based on a formalization by the authors of *GD* [1].

To get an intuition of the ideas of *GA*, consider again the definition  $L \equiv \neg L$  and how it behaves in *GA*, after already having discussed how it behaves in a classical logic. The definition  $L \equiv \neg L$  is perfectly valid in *GA*. However, when trying to assign a truth value to it, i.e. to prove either  $L$  or  $\neg L$ , it is quickly apparent that this is not possible using the *GA* inference rules. For example, the derived contradiction rule in *GA* provides no help, as opposed to the classical version. The reason for this is an additional premise of  $p \vee \neg p$ , a circular proof obligation, since it asks for the very truth value assignment we are currently trying to prove. The truth value of  $L$  is not *grounded* in anything.

$$\frac{\Gamma \vdash p \quad \Gamma \cup \{\neg p\} \vdash q \quad \Gamma \cup \{\neg p\} \vdash \neg q}{\Gamma \vdash p}$$

Similarly, many other inference rules in *GA* demand a *grounding* of the involved values as additional premises compared to their classical counterparts. The authors refer to these premises as *habeas quid* conditions.

There is an ongoing formalization project of *GD/GA* in the proof assistant Isabelle/HOL, which already yielded a consistency proof of the quantifier-free fragment of GD, showing great promise for GD as a reasoning framework. However, the other aim of GD is to show that it is also expressive and importantly, usable as a tool for formalizing mathematics itself. It is not clear whether *grounded* reasoning is feasible when aiming to formalize even basic arithmetic. The *GA* formalization in the mature HOL logic enables studying meta-logical properties such as consistency. However, it is not suitable for providing *GD/GA* as a tool for formal reasoning itself for a few reasons.

- Formalizing GD within a mature metalogic such as HOL adds the axioms of the metalogic to the trusted base of GD, which is undesirable from a meta-logical perspective.
- The logical primitives and axioms being embedded within the primitives of another logic (HOL in this case) makes reasoning within it contrived and needlessly complicated.
- A logic is developed largely for idealistic reasons; the authors believe its reasoning principles are the right ones for at least some domain. Formalizing such a logic within another rich logic means that its reasoning principles are simply embedded in the, likely very different principles, of the meta-logic, defeating that purpose.

It is thus highly desirable to formalize a foundational formal system like GD atop a very minimal reasoning framework.

This is exactly what Isabelle provides with the Pure framework: A minimal, generic logical calculus to formalize object logics on top of. Any object logic in Isabelle, including Isabelle/HOL, is formalized atop Pure.

---

This thesis aims to fully axiomatize GA in Pure, yielding essentially an interactive theorem prover Isabelle/GA, which can be used for formal reasoning based directly on the reasoning principles and axioms of GA. The next goal is to formalize large chunks of basic arithmetic to evaluate the feasibility of GA as a foundation of mathematical reasoning and provide tooling and proof automation to try and make *grounded* reasoning as small of an inconvenience over classical reasoning as possible.

# Background

2

## 2.1 Isabelle/Pure

Isabelle provides a logical framework called *Pure*. It contains a minimal meta-logic, which is a typed lambda calculus with few additional connectives, some keywords to add types and constants to said calculus, and a structured proof language called Isar. Any object logic in Isabelle, for example the highly mature Isabelle/HOL fragment, are formalized atop *Pure*.

Isabelle itself is implemented in the Standard ML (SML) programming language, and implementing an object logic *Pure* almost always requires writing SML for things like proof automation, providing keywords, methods, or definitional mechanisms for users, and various other tooling such as code extraction.

This subsection provides a formalization of the *Pure* calculus.

Unfortunately, there is no single document that lays out the syntax, axioms, and derivation rules of the *Pure* calculus in their entirety. The following is an attempt at providing such a characterization, combining information from two Isabelle papers [2], [3] and the Isabelle reference manual [4].

### 2.1.1 Syntax of Pure

The core syntax of *Pure* is a typed lambda calculus, augmented with type variables, universal quantification, equality, and implication.

Propositions are terms of the distinct type `prop`. Propositions in *Pure* are thus terms and not types, like they are in type-theory based provers like Rocq or Lean.

#### Type Syntax

$\tau ::= \alpha$	type variable
$\tau \Rightarrow \tau$	function type
<code>prop</code>	type of propositions

#### Term Syntax

$t ::= x$	variable
$c$	constant
$t\ t$	application
$\lambda x :: \tau. t$	lambda abstraction
$t \Rightarrow t$	implication
$t \equiv t$	equality
$\bigwedge x :: \tau. t$	universal quantification

The symbols used for implication, equality, and universal quantification are non-standard to leave the standard symbols free for object logics.

Even though *Pure* has type variables, it provides no construct to capture them as an argument, and thus also has no for-all type like the polymorphic lambda calculus System F.

### 2.1.2 Equality, Implication, and Quantification as Type Constructors

The connectives equality, implication, and universal quantification are all type constructors of the `prop` type with the following polymorphic type signatures.

$$\begin{aligned} \equiv &:: \alpha \Rightarrow \alpha \Rightarrow \text{prop} \\ \Rightarrow &:: \text{prop} \Rightarrow \text{prop} \Rightarrow \text{prop} \\ \bigwedge &:: (\alpha \Rightarrow \text{prop}) \Rightarrow \text{prop} \end{aligned}$$

The arguments of  $\equiv$  are the two operands to compare, the arguments for  $\Rightarrow$  are the sequent and consequent respectively, while the argument of  $\bigwedge$  is a function from the type whose inhabitants are quantified over to the term that represents the body of the quantifier.

Since type variables denote only a single, albeit arbitrary, type, there is technically one instance of each polymorphic connective for every given type. For example, for any type  $\sigma$ , there is a constant  $\equiv_{\sigma} :: \sigma \Rightarrow \sigma \Rightarrow \text{prop}$ .

### 2.1.3 Deduction Rules

The operational semantics of the underlying lambda calculus and its typing rules are standard and thus omitted. The following discusses the more interesting deduction rules, which make *Pure* a logical framework.

Relative to an object logic with a set of defined axioms  $A$  any axiom  $\alpha \in A$  can always be derived, as can any assumption  $\gamma \in \Gamma$ .

#### Basic Rules

$$\frac{A \text{ axiom}}{\Gamma \vdash A} \quad (\text{Axiom}) \qquad \frac{A \in \Gamma}{\Gamma \vdash A} \quad (\text{Ass})$$

The implication and universal quantification introduction and elimination rules are standard.

#### Implication Deduction Rules

$$\frac{\Gamma \cup \{A\} \vdash B}{\Gamma \vdash A \Rightarrow B} \quad (\Rightarrow I) \qquad \frac{\Gamma_1 \vdash A \Rightarrow B \quad \Gamma_2 \vdash A}{\Gamma_1 \cup \Gamma_2 \vdash B} \quad (\Rightarrow E)$$

#### Universal Quantification Deduction Rules



$$\frac{\Gamma \vdash B(x) \quad x \text{ not free in } \Gamma}{\Gamma \vdash \bigwedge x.B(x)} \quad (\bigwedge I) \qquad \frac{\Gamma \vdash \bigwedge x.B(x)}{\Gamma \vdash B(a)} \quad (\bigwedge E)$$

For equality, besides the expected deduction rules corresponding to the equivalence relation properties, there are also deduction rules for equality of lambda abstractions and **prop**, the latter of which is defined as equivalence of truth values ( $a \Rightarrow b$  and  $b \Rightarrow a$ ).

#### Equality Deduction Rules

$$\frac{}{\Gamma \vdash a \equiv a} \quad (\equiv \text{Refl}) \quad \frac{\Gamma \vdash b \equiv a}{\Gamma \vdash a \equiv b} \quad (\equiv \text{Sym}) \quad \frac{\Gamma \vdash a \equiv b \quad \Gamma \vdash b \equiv c}{\Gamma \vdash a \equiv c} \quad (\equiv \text{Trans})$$

$$\frac{\Gamma \vdash a \equiv b}{\Gamma \vdash (\lambda x.a) \equiv (\lambda x.b)} \quad (\equiv \text{Lam}) \quad \frac{\Gamma \vdash a \Rightarrow b \quad \Gamma \vdash b \Rightarrow a}{\Gamma \vdash a \equiv b} \quad (\equiv \text{prop})$$

The  $\lambda$ -conversion rules facilitate equivalence reasoning for lambda abstractions. The rules are  $\alpha$ -conversion,  $\beta$ -conversion and extensionality. The notation  $a[y/x]$  expresses the substitution of  $x$  with  $y$  in  $a$ , that is, all occurrences of  $x$  in  $a$  are replaced with  $y$ .

#### Lambda Conversion Rules

$$\frac{y \text{ not free in } a}{\Gamma \vdash (\lambda x.a) \equiv (\lambda y.a[y/x])} \quad (\alpha\text{-Conv}) \qquad \frac{}{\Gamma \vdash (\lambda x.a) b \equiv a[b/x]} \quad (\beta\text{-Conv})$$

$$\frac{\Gamma \vdash f x \equiv g x \quad x \text{ not free in } \Gamma, f, \text{ and } g}{\Gamma \vdash f \equiv g} \quad (\text{Ext})$$

Finally, the equivalence substitution rule:

#### Equivalence Elimination

$$\frac{\Gamma \vdash a \equiv b \quad \Gamma \vdash a}{\Gamma \vdash b} \quad (\equiv E)$$

### 2.1.4 Formalizing Object Logics in Pure

An object logic in *Pure* is created by adding new types, constants and axioms. That is, the *Pure* logic is extended.

It is convention to define a new propositional type in an object logic, which is used as the type of propositions in the *object* logic, as opposed to the *meta* logic, which is *Pure*.

This is achieved using the `typedcl` keyword, which declares a syntactic type in the *Pure* calculus. This type has no known inhabitants or any other information yet.

typed decl  $o$

Any information about  $o$  must be axiomatized. For example, the following declares typed constants `disj` and `True` and axiomatizes certain rules about them.

```
1 axiomatization
2   True :: <o> and
3   disj :: <o  $\Rightarrow$  o  $\Rightarrow$  o> (infixr <v> 30)
4 where
5   true:   <True>
6   disjI1: <P  $\Rightarrow$  P v Q> and
7   disjI2: <Q  $\Rightarrow$  P v Q> and
```

Isabelle

The axiomatized rules here simply state that `True` holds and that from either  $P$  or  $Q$ ,  $P \vee Q$  can be derived. Here,  $P$  and  $Q$  are implicitly universally quantified, ranging over all terms of type `prop`. That is,  $P$  and  $Q$  can be substituted for any term of the correct type (which is  $o$  for both  $P$  and  $Q$  here). Now, the type  $o$  has known inhabitants and structure. However, Isabelle (or rather, *Pure*) cannot reason about it, because it cannot connect the type  $o$  meaningfully with its meta-theory. To resolve this, a judgment must translate from the object-level proposition type  $o$  to the meta-level type `prop`.

```
1 judgment
2   Trueprop :: <o  $\Rightarrow$  prop> (<_> 5)
```

Isabelle

The syntax annotation  $(\langle\_ \rangle 5)$  means that any term of type  $o$  is implicitly augmented with the `Trueprop` judgment. The very low precedence value of 5 ensures that the `Trueprop` judgment is only applied to top-level terms. For example, the term  $x \vee \text{True}$  is the same as `Trueprop (x  $\vee$  True)` and both are of type `prop` due to the `Trueprop` predicate converting the formula to that type.

As you might have noticed, we have made use of this implicit conversion from  $o$  to `prop` already in the axiomatization block from earlier. That is, the `Trueprop` judgment must be declared before the axiomatization block, else the latter will just report a typing error.

Now, we can state and prove a first lemma in this tiny object logic, using the previously defined axioms.

```
1 lemma "x v True"
2   apply (rule disjI2)
3   apply (rule true)
4   done
```

Isabelle

Applying `disjI2` ‘selects’ the second disjunct to prove, which results in the subgoal `True`, which in turn we can solve using the `true` axiom.

This short introduction suffices for now, as we will later implement a much richer logic, Grounded Deduction, using these same basic constructs. We can clearly see that implementing an object logic in *Pure* actually extends *Pure*, in the sense that it adds new types and deduction rules. For example, our extension added a type and three symbols to the existing syntax of *Pure*. If we call the tiny logic formalized above *Pure'*, the following is its type and term syntax:

**Type Syntax of Pure'**

$\tau ::= \alpha$	type variable
$\tau \Rightarrow \tau$	function type
<b>prop</b>	type of propositions
$\mathbf{o}$	type of object logic propositions

**Term Syntax of Pure'**

$t ::= x$	variable
$c$	constant
$t\ t$	application
$\lambda x :: \tau. t$	lambda abstraction
$t \Rightarrow t$	implication
$t \equiv t$	equality
$\bigwedge x :: \tau. t$	universal quantification
<b>True</b>	$\mathbf{o}$ -typed true constant
$t \vee t$	$\mathbf{o}$ -typed logical or
<b>Trueprop</b> $t$	conversion function $\mathbf{o}$ to <b>prop</b>

Further, we can view the added axioms as new inference rules, with the explicit **Trueprop** function application.

$$\frac{\Gamma \vdash \text{Trueprop } P}{\Gamma \vdash \text{Trueprop } P \vee Q} \quad (\text{disjI1}) \qquad \frac{\Gamma \vdash \text{Trueprop } Q}{\Gamma \vdash \text{Trueprop } P \vee Q} \quad (\text{disjI2})$$

$$\frac{}{\Gamma \vdash \text{Trueprop } \text{True}} \quad (\text{true})$$

It is technically possible to avoid declaring a new proposition type for an object logic and instead use **prop** directly as the type of propositions. However, doing so means that the (object) logic immediately inherits the built-in connectives and deduction rules, such as implication ( $\Rightarrow$ ) and universal quantification ( $\bigwedge$ ), and the sequent-style reasoning built into the kernel.

Such a structure reduces the control one has over the logic and keeps many reasoning principles implicit.

## 2.2 Grounded Arithmetic (GA)

This subsection provides a full characterization of GA, a first-order formalization of arithmetic based on the principles of GD. This is the fragment that is later formalized in Isabelle.

GA makes definitions first-class objects in the logic and allows arbitrary references of the symbol currently being defined or other, previously defined symbols, in the expanded term.

To prevent immediate inconsistency, GA must weaken other deduction rules commonly seen in classical logic. Specifically, GA adds a so-called *habeas quid* sequent to many inference rules. Intuitively, this means that in certain inference rules, a (sub)term must first be shown to terminate.

### 2.2.1 BGA Formalization

We start by formalizing the syntax and axioms of *Basic Grounded Arithmetic* (*BGA*), the quantifier-free fragment of GA, based on the formalization in [1]. This formulation later adds quantifiers by encoding them as unbounded computations in *BGA*, yielding full *GA*. This however requires a sophisticated encoding using Gödel-style reflection, i.e. encoding its own term syntax into natural numbers, which is out of scope for a formalization in *Pure*. Thus, we will later add quantifiers by simply axiomatizing them.

The primitive term syntax of BGA is the following.

#### BGA Primitive Term Syntax

$t ::= x$	variable
0	natural-number constant zero
$\mathbf{S}(t)$	natural-number successor
$\mathbf{P}(t)$	natural-number predecessor
$\neg t$	logical negation
$t \vee t$	logical disjunction
$t = t$	natural-number equality
if $t$ then $t$ else $t$	conditional evaluation
$d(t, \dots, t)$	application of recursive definition

It is noteworthy that the GA term syntax mixes expressions that are natural numbers and expressions that are formulas into the same syntactic category. For example, the expression  $S(x) = x \vee x$  is a valid term according to the syntax, despite the left-hand side shape clearly indicating a natural number, while the right hand side shape indicates a truth value.

Besides the primitives, other constants and logical connectives are defined as notational shorthands using the primitives.

#### Notational Shorthands

$\text{True} \equiv 0 = 0$	true constant
$\text{False} \equiv 0 = S(0)$	false constant
$a \mathbf{N} \equiv a = a$	number type
$p \mathbf{B} \equiv p \vee \neg p$	boolean type
$p \wedge q \equiv \neg(\neg p \vee \neg q)$	logical conjunction
$p \rightarrow q \equiv \neg p \vee q$	implication
$p \leftrightarrow q \equiv (p \rightarrow q) \wedge (q \rightarrow p)$	biconditional
$a \neq b \equiv \neg(a = b)$	inequality

The surprising shorthands are  $a \mathbf{N}$  and  $p \mathbf{B}$ . The latter is a predicate over  $p$  deciding whether it is a binary truth value. In a logic with the law of excluded middle,  $p \mathbf{B}$  would be a tautology for any  $p$ , but in a logic without it, it can be interpreted as a termination certificate for truth values. Similarly,  $a \mathbf{N}$  can be interpreted as a termination certificate for natural number expressions. The shorthand itself is surprising, because if equality is reflexive,  $a = a$  is true for any  $a$ . In GA however, equality is not reflexive as we will soon see, and a proof of  $a = a$  is equivalent to a termination proof of the expression.

The syntax does not mean much without a set of axioms giving them meaning. We start with listing the propositional logic axioms.

In the following,  $\Gamma$  denotes a set of background assumptions. For completeness sake, the explicit structural rules governing this set of assumptions is listed here. Since  $\Gamma$  is a set, the usually explicit rules for permuting and duplicating assumptions are not needed.

### Structural Rules

$$\frac{}{\Gamma \cup \{p\} \vdash p} \quad (\text{H}) \qquad \frac{\Gamma \vdash q}{\Gamma \cup \{p\} \vdash q} \quad (\text{W})$$

### Propositional Logic Axioms

$$\begin{array}{lll} \frac{\Gamma \vdash p}{\Gamma \vdash p \vee q} \quad (\vee \text{ I1}) & \frac{\Gamma \vdash q}{\Gamma \vdash p \vee q} \quad (\vee \text{ I2}) & \frac{\Gamma \vdash \neg p \quad \Gamma \vdash \neg q}{\Gamma \vdash \neg(p \vee q)} \quad (\vee \text{ I3}) \\ \\ \frac{\Gamma \vdash p}{\Gamma \vdash \neg \neg p} \quad (\neg \neg \text{ IE}) & \frac{\Gamma \vdash p \quad \Gamma \vdash \neg p}{\Gamma \vdash q} \quad (\neg E) & \frac{\Gamma \vdash \neg(p \vee q)}{\Gamma \vdash \neg p} \quad (\vee \text{ E1}) \\ \\ \frac{\Gamma \vdash \neg(p \vee q)}{\Gamma \vdash \neg q} \quad (\vee \text{ E2}) & \frac{\Gamma \vdash p \vee q \quad \Gamma \cup \{p\} \vdash r \quad \Gamma \cup \{q\} \vdash r}{\Gamma \vdash r} \quad (\vee \text{ E3}) \end{array}$$

Rules such as  $\neg \neg \text{ IE}$  with a double-line are bidirectional, i.e. they serve as both an introduction and elimination rule.

The propositional axioms are fairly standard, but inclusion of double negation elimination is notable, as this is common in classical logics, but omitted in computational logics.

**Equality Axioms**

$$\frac{\Gamma \vdash a = b}{\Gamma \vdash b = a} \quad (= S) \qquad \frac{\Gamma \vdash a = b \quad \Gamma \vdash K a}{\Gamma \vdash K b} \quad (= E)$$

The equality axioms notably omit reflexivity. Symmetry of equality is an axiom, as is equality substitution in an arbitrary context  $K$ . Transitivity of equality can be deduced using equality substitution.

**Natural Number Axioms**

$$\begin{array}{lll} \frac{}{\Gamma \vdash 0 \mathbf{N}} \quad (0I) & \frac{\Gamma \vdash a = b}{\Gamma \vdash \mathbf{S}(a) = \mathbf{S}(b)} \quad (\mathbf{S} = IE) & \frac{\Gamma \vdash a = b}{\Gamma \vdash \mathbf{P}(a) = \mathbf{P}(b)} \quad (\mathbf{P} = I) \\ \\ \frac{\Gamma \vdash a \mathbf{N}}{\Gamma \vdash \mathbf{P}(\mathbf{S}(a)) = a} \quad (\mathbf{P} = I2) & \frac{\Gamma \vdash a \mathbf{N}}{\Gamma \vdash \mathbf{S}(a) \neq 0} \quad (\mathbf{S} \neq 0I) & \frac{\Gamma \vdash c \quad \Gamma \vdash a \mathbf{N}}{\Gamma \vdash (\text{if } c \text{ then } a \text{ else } b) = a} \quad (? I1) \\ \\ \frac{\Gamma \vdash \neg c \quad \Gamma \vdash b \mathbf{N}}{\Gamma \vdash (\text{if } c \text{ then } a \text{ else } b) = b} \quad (? I2) & \frac{\Gamma \vdash K 0 \quad \Gamma \cup \{x \mathbf{N}, K x\} \vdash K \mathbf{S}(x) \quad \Gamma \vdash a \mathbf{N}}{\Gamma \vdash K a} \quad (\text{Ind}) & \\ \\ \frac{\Gamma \vdash a \mathbf{N} \quad \Gamma \vdash b \mathbf{N}}{\Gamma \vdash a = b \mathbf{N}} \quad (= \text{TI}) & \frac{\Gamma \vdash c \mathbf{B} \quad \Gamma \vdash a \mathbf{N} \quad \Gamma \vdash b \mathbf{N}}{\Gamma \vdash \text{if } c \text{ then } a \text{ else } b \mathbf{N}} \quad (? \text{TI}) & \end{array}$$

The natural number axioms are fairly close to the standard Peano axioms, with some notable exceptions.

The *grounding* equality is the  $0I$  axiom, postulating that  $0 \mathbf{N}$ , or, by unfolding the definition,  $0 = 0$ . Using the  $\mathbf{S} = IE$  axiom,  $\mathbf{S}(a) \mathbf{N}$  can be deduced for any  $a$  for which  $a \mathbf{N}$  is already known. The induction axiom  $\text{ind}$  has an additional premise of  $a \mathbf{N}$ , i.e. it requires proof that the expression induction is performed over is indeed a (terminating) natural number.

Conditional evaluation is a primitive in  $GA$  and its behavior must thus be axiomatized. The two inference rules correspond to the positive and negative evaluation of the condition, and they both require that the expression from the corresponding branch is shown to be terminating (i.e.  $a \mathbf{N}$  and  $b \mathbf{N}$  respectively). This additional premise prevents equalities of potentially non-terminating expressions to be deduced.

**GROUNDING CONTRADICTION** Although  $GA$  is not classical, a contradiction rule can be derived. The resulting inference rule has an additional  $p \mathbf{B}$  premise not present in the classical version, which demands  $p$  is first shown to have a truth value. To get a feeling for the logic, we construct the proof explicitly in a natural deduction style derivation tree.

**Theorem 1.** *Grounded Contradiction*

$$\frac{\Gamma \vdash p \mathbf{B} \quad \Gamma \cup \{\neg p\} \vdash q \quad \Gamma \cup \{\neg p\} \vdash \neg q}{\Gamma \vdash p}$$

**Proof.**

$$\frac{\frac{\Gamma \vdash p \quad \mathbf{B}}{\Gamma \vdash p \vee \neg p} \text{B def} \quad \frac{}{\Gamma \cup \{p\} \vdash p} \text{H} \quad \frac{\Gamma \cup \{\neg p\} \vdash q \quad \Gamma \cup \{\neg p\} \vdash \neg q}{\Gamma \cup \{\neg p\} \vdash p} \neg E}{\Gamma \vdash p} \vee E3$$

■

**GROUNDING IMPLICATION** Implication is not a primitive in  $GA$ , but rather the shorthand  $a \rightarrow b \equiv \neg a \vee b$ . From this definition, the classical elimination rule *modus ponens* can be derived. However, only a weakened introduction rule, with the now familiar additional *habeas quid* premise, can be derived.

**Theorem 2.** *Modus Ponens*

$$\frac{\Gamma \vdash p \quad \Gamma \vdash p \rightarrow q}{\Gamma \vdash q} (\rightarrow E)$$

**Proof.**

$$\frac{\frac{}{\Gamma \cup \{q\} \vdash q} \text{H} \quad \frac{\frac{}{\Gamma \cup \{\neg p\} \vdash \neg p} \text{H} \quad \frac{\Gamma \vdash p}{\Gamma \cup \{\neg p\} \vdash p} \text{W}}{\Gamma \cup \{\neg p\} \vdash q} \neg E \quad \frac{\Gamma \vdash p \rightarrow q}{\Gamma \vdash \neg p \vee q} \rightarrow \text{def}}{\Gamma \vdash q} \vee E3$$

■

**Theorem 3.** *Implication Introduction*

$$\frac{\Gamma \vdash p \quad \mathbf{B} \quad \Gamma \cup \{p\} \vdash q}{\Gamma \vdash p \rightarrow q} (\rightarrow I)$$

**Proof.**

$$\frac{\frac{\Gamma \vdash p \quad \mathbf{B}}{\Gamma \vdash p \vee \neg p} \text{B def} \quad \frac{\Gamma \cup \{p\} \vdash q}{\Gamma \cup \{p\} \vdash \neg p \vee q} \vee E2 \quad \frac{\frac{}{\Gamma \cup \{\neg p\} \vdash \neg p} \text{H}}{\Gamma \cup \{\neg p\} \vdash \neg p \vee q} \vee I1}{\Gamma \vdash \neg p \vee q} \vee E3}{\Gamma \vdash p \rightarrow q} \rightarrow \text{def}$$

■

**DEFINITIONAL AXIOMS** Finally, the axioms for definitions allow arbitrary substitution of a symbol with its definition body (and the other way around) in any context. The vector notation  $\vec{a}$  denotes an argument vector for the defined function symbol.

**Definition Axioms**

$$\frac{\Gamma \vdash s(\vec{x}) \equiv d(\vec{x}) \quad \Gamma \vdash K d(\vec{a})}{\Gamma \vdash K s(\vec{a})} (\equiv E) \quad \frac{\Gamma \vdash s(\vec{x}) \equiv d(\vec{x}) \quad \Gamma \vdash K s(\vec{a})}{\Gamma \vdash K d(\vec{a})} (\equiv I)$$

### 2.2.2 GA with Axiomatized Quantifiers

As already mentioned, the creators of *GA* claim that quantifiers can be encoded into BGA using the powerful definitional mechanism [1], yielding full *GA* “for free”. However, as this will not be feasible in the formalization within *Pure*, the following axiomatizes the quantifiers instead. The axioms correspond to the inference rules that would be derivable from encoded quantifiers [1].

#### Quantifier Axioms

$$\begin{array}{ll}
 \frac{\Gamma \cup \{x \mathbf{N}\} \vdash K \ x}{\Gamma \vdash \forall x. K \ x} \quad (\forall I) & \frac{\Gamma \vdash \forall x. K \ x \quad \Gamma \vdash a \ \mathbf{N}}{\Gamma \vdash K \ a} \quad (\forall E) \\
 \frac{\Gamma \vdash a \ \mathbf{N} \quad \Gamma \vdash K \ a}{\Gamma \vdash \exists x. K \ x} \quad (\exists I) & \frac{\Gamma \vdash \exists x. K \ x \quad \Gamma \cup \{x \ \mathbf{N}, K \ x\} \vdash q}{\Gamma \vdash q} \quad (\exists E)
 \end{array}$$

Besides the additional *habeas quid* premises, the quantifier axioms are standard.

Since the quantifiers are primitive here, they must be added to the primitive term syntax, yielding the full *GA* primitive term syntax.

#### GA Primitive Term Syntax

$t ::= x$	variable
0	natural-number constant zero
$\mathbf{S}(t)$	natural-number successor
$\mathbf{P}(t)$	natural-number predecessor
$\neg t$	logical negation
$t \vee t$	logical disjunction
$t = t$	natural-number equality
if $t$ then $t$ else $t$	conditional evaluation
$d(t, \dots, t)$	application of recursive definition
$\forall x. t$	universal quantifier
$\exists x. t$	existential quantifier

This set of axioms is now a full formalization of a grounded flavor of first-order arithmetic, which we just refer to as *GA* from now on.



# Formalizing GA in Pure

3

This chapter aims to translate the full formalization of *GA* in Section 2.2 into Isabelle/Pure. Formally, this means that *GA* is embedded into the *Pure* calculus, using it as a meta-logic. The typed lambda calculus *Pure* is fully characterized in Section 2.1. In the following, all types mentioned are part of the *Pure* type system. *GA* itself has no real (conventional) notion of types, although the ‘inherited’ types of *Pure* can also be interpreted as syntactic *GA* types.

## 3.1 Proposotional Axioms

We first declare a syntactic Isabelle type for truth values in *GA* `o` and a function to convert from `o` to the type of truth values of the *Pure* calculus `prop`, as explained in Section 2.1.4.

```
1  typedecl o
2
3  judgment
4  Trueprop :: <o ⇒ prop>  (<_> 5)
```

Isabelle

Now, we can declare constants `disj` and `not`, with the propositional axioms from Section 2.2.1.

```
1  axiomatization
2    disj :: <o ⇒ o ⇒ o>  (infixr <v> 30) and
3    not  :: <o ⇒ o>  (<¬ _> [40] 40)
4  where
5    disjI1: <P ⇒ P v Q> and
6    disjI2: <Q ⇒ P v Q> and
7    disjI3: <[[¬P; ¬Q]] ⇒ ¬(P v Q)> and
8    disjE1: <[[P v Q; P ⇒ R; Q ⇒ R]] ⇒ R> and
9    disjE2: <¬(P v Q) ⇒ ¬P> and
10   disjE3: <¬(P v Q) ⇒ ¬Q> and
11   dNegI: <P ⇒ (¬¬P)> and
12   dNegE: <(¬¬P) ⇒ P> and
13   exF: <[[P; ¬P]] ⇒ Q>
```

Isabelle

This immediately introduces the infix notation  $a \vee b$  for `disj a b` and  $\neg a$  for `not a`, where `infixr` means that the provided syntax is an infix operator that associates to the right. The precedence of the  $\neg$  operator is 40, it thus binds stronger than the  $\vee$  operator with precedence 30.

In general, a rule in the natural deduction style formalization from Section 2.2 of the following shape:

$$\frac{\Gamma \cup A_1 \vdash P_1 \quad \Gamma \cup A_2 \vdash P_2 \quad \dots \quad \Gamma \cup A_n \vdash P_n}{\Gamma \vdash C}$$

Is translated into the following shape in Isabelle:

```
1 (A1  $\Rightarrow$  P1)  $\Rightarrow$  (A2  $\Rightarrow$  P2)  $\Rightarrow$  ...  $\Rightarrow$  (An  $\Rightarrow$  Pn)  $\Rightarrow$  C
```

Isabelle

Or, equivalently:

```
1  $\llbracket$ A1  $\Rightarrow$  P1; A2  $\Rightarrow$  P2; ...; An  $\Rightarrow$  Pn $\rrbracket \Rightarrow$  C
```

Isabelle

Potential background assumptions  $\Gamma$  do not need to be explicitly managed in Isabelle. What we consider a derivation, i.e.  $\vdash$  in *GA* is just implication  $\Rightarrow$  in *Pure*.

Theorems in this section that are explicitly presented in a *theorem block* keep using the natural deduction style to be consistent with Section 2.1 and Section 2.2. In this notation, deducability from assumptions is denoted  $\vdash$ , while in Isabelle the corresponding symbol is  $\Rightarrow$ , i.e. meta-level implication.

### 3.2 Natural Number Axioms

We declare a type `num` for natural numbers.

```
1 typedef num
```

Isabelle

Before the majority of natural number axioms, we define equality and some connectives derived from the primitives. We axiomatize equality with unrestricted substitution and symmetry. Equality is not just defined for `num`, but for any type '`a`', where '`a`' is a generic type variable. As expected, equality is a binary infix operator associating to the left.

```
1 axiomatization
2   eq :: <'a  $\Rightarrow$  'a  $\Rightarrow$  o> (infixl <=> 45)
3 where
4   eqSubst: < $\llbracket$ a = b; Q a $\rrbracket \Rightarrow$  Q b> and
5   eqSym: <a = b  $\Rightarrow$  b = a>
```

Isabelle

Transitivity can be proven using the substitution axiom.

```
1 lemma eq_trans: "a = b  $\Rightarrow$  b = c  $\Rightarrow$  a = c"
2 by (rule eqSubst[where a="b" and b="c"], assumption)
```

Isabelle

However, transitivity is not a very useful lemma, as it just as efficient to directly use equality substitution.

Inequality, the **B** and **N** judgments, and other logical operators can now be defined in terms of the axiomatized primitives. Notably, **B** is explicitly typed at  $o \Rightarrow o$ , meaning that e.g. the term **S**(zero) **B** is rejected by the Isabelle parser, as it is not well-typed. Similarly, **N** is explicitly typed at  $\text{num} \Rightarrow o$ , only accepting an argument of type `num`.

```
1 definition neq :: <num  $\Rightarrow$  num  $\Rightarrow$  o> (infixl < $\neq$ > 45)
2   where <a  $\neq$  b  $\equiv$   $\neg$  (a = b)>
3 definition bJudg :: <o  $\Rightarrow$  o> (<_ B> [21] 20)
4   where <(p B)  $\equiv$  (p  $\vee$   $\neg$ p)>
5 definition isNat :: <num  $\Rightarrow$  o> (<_ N> [21] 20)
```

Isabelle

```

6  where "x N  $\equiv$  x = x"
7  definition conj :: <o  $\Rightarrow$  o  $\Rightarrow$  o> (infixl < $\wedge$ > 35)
8    where <p  $\wedge$  q  $\equiv \neg(\neg p \vee \neg q)$ >
9  definition impl :: <o  $\Rightarrow$  o  $\Rightarrow$  o> (infixr < $\rightarrow$ > 25)
10   where <p  $\rightarrow$  q  $\equiv \neg p \vee q$ >
11 definition iff :: <o  $\Rightarrow$  o  $\Rightarrow$  o> (infixl < $\leftrightarrow$ > 25)
12   where <p  $\leftrightarrow$  q  $\equiv (p \rightarrow q) \wedge (q \rightarrow p)$ >

```

As an example, all the introduction and elimination rules for conjunction  $\wedge$  can be proven now:

```

1  lemma conjE1:
2    assumes p_and_q: "p  $\wedge$  q"
3    shows "p"
4    apply (rule dNegE)
5    apply (rule disjE2[where Q=" $\neg q$ "])
6    apply (fold conj_def)
7    apply (rule p_and_q)
8    done
9
10 lemma conjE2:
11   assumes p_and_q: "p  $\wedge$  q"
12   shows "q"
13   apply (rule dNegE)
14   apply (rule disjE3[where P=" $\neg p$ "])
15   apply (fold conj_def)
16   apply (rule p_and_q)
17   done
18
19 lemma conjI:
20   assumes p: "p"
21   assumes q: "q"
22   shows "p  $\wedge$  q"
23   apply (unfold conj_def)
24   apply (rule disjI3)
25   apply (rule dNegI)
26   apply (rule p)
27   apply (rule dNegI)
28   apply (rule q)
29   done

```

Isabelle

We can now axiomatize the zero constant and `pred` and `suc` functions. Compared to the formalization in Section 2.2, there is an additional axiom here stating  $P(0) = 0$ . This will allow stating some convenient lemmas later, for example that for any  $n$  with  $n \in \mathbf{N}$ ,  $P(n) \leq n$ .

```

1  Isabelle

```

```

2  axiomatization
3    zero :: <num>                                and
4    suc  :: <num ⇒ num>    (<S(_)> [800]) and
5    pred :: <num ⇒ num>    (<P(_)> [800])
6  where
7    nat0: <zero N> and
8    sucInj: <S a = S b ⇒ a = b> and
9    sucCong: <a = b ⇒ S a = S b> and
10   predCong: <a = b ⇒ P a = P b> and
11   eqBool: <[[a N; b N]] ⇒ (a = b) B> and
12   eqBoolB: <[[x B; y B]] ⇒ (x = y) B> and
13   sucNonZero: <a N ⇒ S a ≠ zero> and
14   predSucInv: <a N ⇒ P(S(a)) = a> and
15   pred0: <P(zero) = zero> and
16   ind: <[[a N; Q zero; ∧x. x N ⇒ Q x ⇒ Q S(x)]] ⇒ Q a>

```

The following two crucial inference rules can be proven from these axioms, whose proof is given directly in the Isabelle formalization.

**Theorem 4.** *Natural Number Typing Rules*

$$\frac{\Gamma \vdash a \text{ N}}{\Gamma \vdash S(a) \text{ N}} \quad (\text{S N}) \qquad \frac{\Gamma \vdash a \text{ N}}{\Gamma \vdash P(a) \text{ N}} \quad (\text{P N})$$

**Proof.**

```

1  lemma natS:
2    assumes a_nat: "a N"
3    shows "S a N"
4  apply (unfold isNat_def)
5  apply (rule sucCong)
6  apply (fold isNat_def)
7  apply (rule a_nat)
8  done
9
10 lemma natP:
11   assumes a_nat: "a N"
12   shows "P a N"
13 apply (unfold isNat_def)
14 apply (rule predCong)
15 apply (fold isNat_def)
16 apply (rule a_nat)
17 done

```

Isabelle



The constants True and False can now be defined as canonical truth and falsehood  $\text{True} \equiv (0 = 0)$  and  $\text{False} \equiv (S(0) = 0)$ .

```
1 definition True
2   where <True  $\equiv$  zero = zero>
3 definition False
4   where <False  $\equiv$  S(zero) = zero>
```

Isabelle

### 3.3 Grounded Contradiction

The grounded contradiction lemma proven as a derivation tree in Section 2.2.1.1 can now be proven in the Isabelle formalization.

#### Grounded Contradiction

$$\frac{\Gamma \vdash p \text{ B} \quad \Gamma \cup \{\neg p\} \vdash q \quad \Gamma \cup \{\neg p\} \vdash \neg q}{\Gamma \vdash p}$$

```
1 lemma grounded_contradiction:
2   assumes p_bool: <p B>
3   assumes notp_q: < $\neg p \Rightarrow q$ >
4   assumes notp_notq: < $\neg p \Rightarrow \neg q$ >
5   shows <p>
6 proof (rule disjE1[where P="p" and Q=" $\neg p$ "])
7   show "p  $\vee$   $\neg p$ "
8     using p_bool unfolding GD.bJudg_def .
9   show "p  $\Rightarrow$  p" by assumption
10  show " $\neg p \Rightarrow$  p"
11  proof -
12    assume not_p: " $\neg p$ "
13    have q: "q" using notp_q[OF not_p] .
14    have not_q: " $\neg q$ " using notp_notq[OF not_p] .
15    from q and not_q show "p"
16      by (rule exF)
17  qed
18 qed
```

Isabelle

### 3.4 Syntax Translation for Natural Numbers

The axiomatized symbols for natural numbers are of the shape  $([SP]^+ \text{ zero})$  as a regex. For example, `zero` and `S(S(P zero))` are natural numbers in this GA formalization. It would be better to use the familiar base 10 system, such that the user can write a base-10 number and it is correctly interpreted as the corresponding `num` expression by Isabelle. Luckily, Isabelle provides powerful syntax translation support.

The following snippet achieves this by first declaring an implicit conversion function from numerical tokens (resulting from user input) to the natural number type `num`. The annotation `("_")` means it is applied to all such tokens implicitly. Then, an SML file called *peano\_numerals.ML* provides the function *parse\_gd\_numeral*, which translates a number in base 10 to the `num` version (for example 3 to  $\mathbf{S}(\mathbf{S}(\mathbf{S}(\text{zero})))$ ).

Finally, the command *parse\_translation* specifies that the previously declared `_gd_num` constant performs the action specified by the *parse\_gd\_numeral* function.

```
1 syntax
2   "_gd_num" :: "num_token => num"      ("_")
3
4 ML_file "peano_numerals.ML"
5
6 parse_translation <
7   [(@{syntax_const "_gd_num"}, Peano_Syntax.parse_gd_numeral)]
8 >
```

Isabelle

As mentioned before, Isabelle is implemented mostly in (a dialect of) Standard ML (SML). The SML infrastructure of Isabelle is not meant to be completely abstracted away from the (advanced) user, but rather does Isabelle provide a rich API of SML functions and types, which is collectively referred to as Isabelle/ML. A syntax translation is precisely the kind of task that can be implemented in SML, hooking into the Isabelle implementation itself.

The following are the contents of the *peano\_numerals.ML* file, providing the translation logic from a string representation of a base-10 number to a `num` expression.

```
1  structure Peano_Syntax = struct
2
3    fun nat_to_peano 0 = Syntax.const @{const_syntax "zero"}
4      | nat_to_peano n = Syntax.const @{const_syntax "suc"} $ nat_to_peano (n -
5        1)
6
7    fun parse_gd_numeral _ [Free (s, _)] =
8      (case Int.fromString s of
9        SOME n => nat_to_peano n
10       | NONE => error ("Not a numeral: " ^ s))
11    | parse_gd_numeral _ _ = error "Unexpected numeral syntax"
12  end
```

SML

The function `Int.fromString` is part of the Isabelle/ML API and tries to convert the input string to an (SML) `Int`. The translation from `Int` is then straightforward – 0 is translated to `zero` and `n` to `S` prepended to the recursive translation of `n-1`.

### 3.5 Quantifier Axiomatization

The quantifier axioms exactly implement the ones from the pen-and-paper formalization in Section 2.2.2, displayed again here for reference.

$$\begin{array}{c}
 \frac{\Gamma \cup \{x \mathbf{N}\} \vdash K \ x}{\Gamma \vdash \forall x. K \ x} \quad (\forall I) \qquad \frac{\Gamma \vdash \forall x. K \ x \quad \Gamma \vdash a \ \mathbf{N}}{\Gamma \vdash K \ a} \quad (\forall E) \\
 \\
 \frac{\Gamma \vdash a \ \mathbf{N} \quad \Gamma \vdash K \ a}{\Gamma \vdash \exists x. K \ x} \quad (\exists I) \qquad \frac{\Gamma \vdash \exists x. K \ x \quad \Gamma \cup \{x \ \mathbf{N}, K \ x\} \vdash q}{\Gamma \vdash q} \quad (\exists E)
 \end{array}$$

A crucial detail is the explicit typing of the quantifiers at the type  $(\text{num} \Rightarrow o) \Rightarrow o$ . It makes the quantifiers range only over `num`, crucial as *GA* aims to be a first-order theory over the natural numbers. In a higher-order theory, the quantifiers would range over any type and be typed at  $(a \Rightarrow o) \Rightarrow o$ , where  $a$  is a generic type variable.

```

1 axiomatization
2   forall :: "(num => o) => o"  (binder "∀" [8] 9) and
3   exists :: "(num => o) => o"  (binder "∃" [8] 9)
4 where
5   forallI: "[[Λx. x N => Q x] = ∀x. Q x]" and
6   forallE: "[[∀c'. Q c'; a N] => Q a]" and
7   existsI: "[[a N; Q a] => ∃x. Q x]" and
8   existsE: "[[∃i. Q i; ∧a. a N => Q a => R] => R]"
    
```

Isabelle

### 3.6 Conditional Evaluation Axiomatization

Conditional evaluation cannot seem to be derived from primitives axiomatized so far and must thus be an axiomatized primitive itself.

The syntax of the conditional operator, with the desired shape of the application `cond a b c` being if  $a$  then  $b$  else  $c$ , seems complicated, but Isabelle is well-equipped to handle syntax like this, with the ability to specify the ‘holes’ in a syntax expression like in the following declaration. This type of notation is called *mifix* in Isabelle, as it mixes infix and prefix notation.

Notably, the two branches of the operator and the return type are typed at the generic  $a$ , indicating that the conditional operator in this formalization is polymorphic. Contrary to the pen-and-paper formalization in Section 2.2.1, this allows for returning truth values in a conditional evaluator (or any other type at that, for example functions of type  $\text{num} \Rightarrow \text{num}$ ), for example the constants `True` and `False`. Although these could be encoded with natural numbers just as well, this requires equality checks of a result and is less elegant. Due to the *habeas quid* premises of the two branching axioms, making the arguments of the syntactic construct generic is not sufficient, it just means that the parser won’t reject a term such as `if c then True else False`. Since neither `True N` nor `False N` can be proven, this term cannot be ‘reduced’ using either `condI1` or `condI2`. Thus, there are three additional axioms for conditional evaluation over values of type `o`, mirroring the ones for conditional evaluation over `num`, but with the *habeas quid* premise of `B` instead of `N`. Now, if `True then True else False = True` can be proven using the axiom `condI1B`.

```

1 consts
    
```

Isabelle

```

2   cond :: <o => 'a => 'a => 'a> (<if _ then _ else _> [25, 24, 24] 24)
3
4   axiomatization where
5   condI1: <[[c; a N]] => (if c then a else b) = a> and
6   condI2: <[[¬c; b N]] => (if c then a else b) = b> and
7   condT: <[[c B; a N; b N]] => if c then a else b N> and
8   condI1B: <[[c; d B]] => (if c then d else e) = d> and
9   condI2B: <[[¬c; e B]] => (if c then d else e) = e> and
10  condTB: <[[c B; d B; e B]] => if c then d else e B>
    
```

### 3.7 Definitional Mechanism Axiomatization

The axioms in the formalization in Section 2.2.1 don't make it entirely clear what the definitional operator with the symbol  $\equiv$  is. It is not part of the primitive syntax, making it meta-logical, in the sense that it is outside the logical calculus. However, it is also a premise in the definitional axioms, implying it behaves like a proposition:

$$\frac{\Gamma \vdash s(\vec{x}) \equiv d(\vec{x}) \quad \Gamma \vdash K d(\vec{a})}{\Gamma \vdash K s(\vec{a})} (\equiv E) \quad \frac{\Gamma \vdash s(\vec{x}) \equiv d(\vec{x}) \quad \Gamma \vdash K s(\vec{a})}{\Gamma \vdash K d(\vec{a})} (\equiv I)$$

The most straightforward implementation of such a mechanism in Isabelle is to just treat it as any other logical connective. Its definition and axiomatization make it conceptually very similar to equality  $=$ .

Like equality, the definitional mechanism is polymorphic, allowing the left-hand side and right-hand side of the definition to be of any (albeit the same) type. The first axiom, called **defE**, is exactly the same as equality substitution, and allows the left-hand side of a definition (the *symbol*), to be substituted by its right-hand side (the *expansion*) in any context. Instead of symmetry, the second equality axiom, the definitional operator has the introduction axiom **defI**, which 'folds' a definition, i.e. replaces the right-hand side of a definition with the left-hand-side in any context.

```

1   axiomatization
2   def :: <'a => 'a => o> (infix <:=> 10)
3   where
4   defE: <[[a := b; Q b]] => Q a> and
5   defI: <[[a := b; Q a]] => Q b>
    
```

Isabelle

This means that a definition  $a := b$  is just another proposition of type  $o$ . For example is the sentence  $3 = 0 \vee L := L$  a well-formed term of type  $o$ , as long as  $L$  is a previously declared constant.

However, with the current set of axioms, a statement of the shape  $a := b$  cannot possibly be proven, as there is no rule that introduces the definitional operator. However, the 'truth' of a definition can be assumed, for example in the following lemma.

```

1   lemma
    
```

Isabelle



```

2   assumes l_def: "l := 0"
3   shows "l = 0"
4   apply (rule defE[OF l_def])
5   apply (fold isNat_def)
6   apply (rule nat0)
7   done

```

The proof first unfolds the definition of  $l$  using the `defE` axiom, yielding the goal state  $0 = 0$ , which can be proven by folding the definition of  $\mathbf{N}$ , resulting in  $0 \mathbf{N}$ , whose truth is postulated by the `nat0` axiom.

To get a ‘globally visible’ definition, the definition must be axiomatized. This does not axiomatize any properties about the defined symbol, it just axiomatizes the ‘equivalence’ of the left-hand side with the right-hand side, which, together with the axioms `defE` and `defI` means they can be substituted for each other in any context. The formalization of *GA* in Isabelle/HOL by the authors includes a consistency proof of the axioms given any fixed finite set of definitions [1]. Thus, axiomatizing definitions maintains consistency, as long as there is only a single definition per symbol.

### 3.8 Defining Arithmetic Functions in GA

Having axiomatized full *GA* in Isabelle/Pure, the next step is to define basic arithmetic functions and prove some lemmas about them.

The recursive definitions of the arithmetic functions are straightforward and correspond to case distinctions over the second argument (of zero and non-zero).

```

1  axiomatization
2    add    :: "num  $\Rightarrow$  num  $\Rightarrow$  num" (infixl "+" 60) and
3    sub    :: "num  $\Rightarrow$  num  $\Rightarrow$  num" (infixl "-" 60) and
4    mult   :: "num  $\Rightarrow$  num  $\Rightarrow$  num" (infixl "*" 70) and
5    div    :: "num  $\Rightarrow$  num  $\Rightarrow$  num" and
6    less   :: "num  $\Rightarrow$  num  $\Rightarrow$  num" (infix "<" 50) and
7    leq    :: "num  $\Rightarrow$  num  $\Rightarrow$  num" (infix " $\leq$ " 50)
8  where
9    add_def: "add x y := if y = 0 then x else S(add x (P y))" and
10   sub_def: "sub x y := if y = 0 then x else P(sub x (P y))" and
11   mult_def: "mult x y := if y = 0 then 0 else (x + mult x (P y))" and
12   leq_def: "leq x y := if x = 0 then 1
13             else if y = 0 then 0
14             else (leq (P x) (P y))" and
15   less_def: "less x y := if y = 0 then 0
16             else if x = 0 then 1
17             else (less (P x) (P y))" and
18   div_def: "div x y := if x < y = 1 then 0 else S(div (x - y) y)"

```

Isabelle

These definitions show the necessity of the predecessor function  $\mathbf{P}$  in  $GA$ . The comparison functions  $\leq$  and  $<$  are defined to compute natural numbers, where 1 encodes truth and 0 falsehood. Note that the division function does not terminate for  $y = 0$ .

The functions  $>$  and  $\geq$  can now be defined in terms of  $<$  and  $\leq$ .

```
1 definition greater :: "num  $\Rightarrow$  num  $\Rightarrow$  num" (infix ">" 50) where
2   "greater x y  $\equiv$  1 - (x  $\leq$  y)"
3
4 definition geq :: "num  $\Rightarrow$  num  $\Rightarrow$  num" (infix " $\geq$ " 50) where
5   "geq x y  $\equiv$  1 - (x < y)"
```

Isabelle

A first property provable about the `less` function is that  $x < 0$  is always false, i.e.  $x < 0 = 0$ .

**Theorem 5.** *Less than zero is false*

$$\overline{\Gamma \vdash x < 0 = 0}$$

**Proof.**

Unfolding the definition `less_def` using the `defE` axiom results in the goal state:

1. (if  $0 = 0$  then 0 else if  $x = 0$  then 1 else  $\mathbf{P}(x) < \mathbf{P}(0)$ ) = 0

As the condition holds, applying the `condI1` axiom results in the goal state:

1.  $0 = 0$
2.  $0 \mathbf{N}$

These can both be discharged easily. The latter is the axiom `nat0` and the former is the unfolded version of this axiom, stated by the lemma `zeroRefl`.

```
1 lemma less_0_false: "(x < 0) = 0"
2   apply (rule defE[OF less_def])
3   apply (rule condI1)
4   apply (rule zeroRefl)
5   apply (rule nat0)
6   done
```

Isabelle

■

### 3.9 Termination Proofs

Due to the *habeas quid* premises of so many axioms, an expression like  $a + b$  becomes truly useful only if  $a + b \mathbf{N}$  is provable. With the interpretation that  $a \mathbf{N}$  is a termination certificate for  $a$ ,  $a \mathbf{N} \Rightarrow b \mathbf{N} \Rightarrow a + b \mathbf{N}$  is essentially a termination proof of the `add` function, conditioned on its operands also being terminating natural numbers themselves.

**Theorem 6.** *Termination of add*

$$\frac{\Gamma \vdash x \text{ N} \quad \Gamma \vdash y \text{ N}}{\Gamma \vdash x + y \text{ N}}$$

**Proof.**

By induction over the second argument.

```

1  lemma add_terminates:
2    assumes x_nat: <x N>
3    assumes y_nat: <y N>
4    shows <add x y N>
5  proof (rule ind[where a=y])
6    show "y N" by (rule y_nat)
7    show "add x 0 N"
8      proof (rule defE[OF add_def])
9        show "if (0 = 0) then x else S(add x P(0)) N"
10       apply (rule eqSubst[where a="x"])
11       apply (rule eqSym)
12       apply (rule condI1)
13       apply (rule zeroRefl)
14       apply (rule x_nat)
15       apply (rule x_nat)
16       done
17     qed
18   show ind_step: "∧a. a N ⇒ ((x + a) N) ⇒ ((x + S(a)) N)"
19   proof (rule defE[OF add_def])
20     fix a
21     assume a_nat: "a N" and BC: "add x a N"
22     show "if (S(a) = 0) then x else S(add x P(S(a))) N"
23     proof (rule condT)
24       show "S(a) = 0 B"
25       apply (rule eqBool)
26       apply (rule natS)
27       apply (rule a_nat)
28       apply (rule nat0)
29       done
30     show "x N" by (rule x_nat)
31     show "S(add x P(S(a))) N"
32     apply (rule GD.natS)
33     apply (rule eqSubst[where a="x+a"])
34     apply (rule eqSubst[where a="a" and b="P(S(a))"])
35     apply (rule eqSym)
36     apply (rule predSucInv)
37     apply (rule a_nat)
38     apply (fold isNat_def)

```

Isabelle

```

39          apply (rule BC)
40          apply (rule BC)
41          done
42      qed
43  qed
44 qed

```

■

Termination proofs of subtraction and multiplication follow the same structure, as they also recurse to the immediate predecessor in the second argument. This recursive structure exactly mirrors induction over the corresponding argument, which is why these proofs are so straightforward, despite spelling them out at the axiom level at this point.

Things get a bit more interesting with the  $\leq$  function, as it recurses in both arguments. The solution is to prove a stronger lemma, which universally quantifies over one argument, and then perform induction over the other argument.

**Theorem 7.** *Termination of leq*

$$\frac{\Gamma \vdash x \text{ N} \quad \Gamma \vdash y \text{ N}}{\Gamma \vdash x \leq y \text{ N}}$$

**Proof.**

By induction over the second argument in the strengthened proposition  $\forall x. x \leq y \text{ N}$ .

```

1  lemma leq_terminates:
2    shows "x N  $\Rightarrow$  y N  $\Rightarrow$  x  $\leq$  y N"
3  proof -
4    have H: "y N  $\Rightarrow$   $\forall x. x \leq y \text{ N}$ "
5    proof (rule ind[where a="y"], simp)
6      show " $\forall x'. x' \leq 0 \text{ N}$ "
7      apply (rule forallI)
8      apply (rule defE[OF leq_def])
9      apply (rule condT)
10     apply (rule eqBool)
11     apply (assumption)
12     apply (rule nat0)
13     apply (rule natS)
14     apply (rule nat0)
15     apply (rule eqSubst[where a="0"])
16     apply (rule eqSym)
17     apply (rule condI1)
18     apply (rule zeroRefl)
19     apply (rule nat0)+
20     done
21  show " $\wedge x. x \text{ N} \Rightarrow (\forall xa. xa \leq x \text{ N}) \Rightarrow (\forall xa. xa \leq S(x) \text{ N})$ "

```

Isabelle

```

22     proof -
23       fix x
24       assume H: "∀xa. xa ≤ x N"
25       show "x N ⇒ ∀xa. xa ≤ S(x) N"
26       proof (rule forallI)
27         fix xa
28         show "x N ⇒ xa N ⇒ xa ≤ S(x) N"
29         apply (rule defE[OF leq_def])
30         apply (rule condT)
31         apply (rule eqBool)
32         apply (assumption)
33         apply (rule nat0)+
34         apply (rule natS, rule nat0)
35         apply (rule condT)
36         apply (rule eqBool)
37         apply (rule natS, assumption)
38         apply (rule nat0)+
39         apply (rule eqSubst[where a="x" and b="P S x"])
40         apply (rule eqSym, rule predSucInv, assumption)
41         apply (rule forallE[where a="P xa"])
42         apply (rule H)
43         apply (rule natP, assumption)
44       done
45     qed
46   qed
47   qed
48   then show "x N ⇒ y N ⇒ x ≤ y N"
49   by (rule forallE)
50 qed

```

■

Although the key ideas of the proof are straightforward – the strengthening of the proposition in line 4 and applying induction over the second argument on line 5 are only one line each – this is clouded by a lot of effort to discharge the *habeas quid* premises and other simple things like replacing a  $\mathbf{P}(\mathbf{S}(x))$  with  $x$  in a subexpression. The latter currently requires an application of equality substitution, an application of equality symmetry, and then applying the `predSucInv` axiom.

This issue is tackled in Section 4, where a lot of proof automation and tactics are introduced to simplify reasoning.

The next goal is to prove termination of the division function. This poses two new challenges.

1. The function does not terminate for  $y = 0$ .
2. The recursive pattern is does not mirror induction, i.e. it does not recurse to the immediate predecessor.

The former is solved relatively easily. One option is to add  $\neg y = 0$  as an explicit assumption to the termination proof. Another, more elegant, option is to restate the theorem in the following way:  $x \mathbf{N} \implies y \mathbf{N} \implies \text{div } x \mathbf{S}(y) \mathbf{N}$ . Now, it holds for any natural numbers  $x$  and  $y$ .

To solve the second problem, a strong induction lemma needs to be proven first. This then allows assuming the induction hypothesis for any  $y' \leq y$  when proving the statement for  $\mathbf{S}(y)$ .

The only difference to the induction axiom is that the hypothesis in the induction step is stronger – instead of  $K x$  it is now  $\bigwedge y. y \leq x = 1 \implies K y$  (in Isabelle notation) or  $y \leq x = 1 \vdash K y$  (in natural deduction style notation).

**Theorem 8. Strong Induction**

$$\frac{\Gamma \vdash a \mathbf{N} \quad \Gamma \vdash K 0 \quad \Gamma \cup \{x \mathbf{N}, \{y \mathbf{N}, y \leq x = 1\} \vdash K y\} \vdash K \mathbf{S}(x)}{\Gamma \vdash K a}$$

**Proof.**

By induction over  $a$  in the strengthened object-level proposition  $\forall x. (x \leq a = 1) \longrightarrow K x$ .

```

1 lemma strong_induction: Isabelle
2   shows "a N  $\Rightarrow$  Q 0  $\Rightarrow$  ( $\wedge x. x \mathbf{N} \Rightarrow$  ( $\wedge y. y \mathbf{N} \Rightarrow y \leq x = 1 \Rightarrow Q y$ )  $\Rightarrow$  ( $Q \mathbf{S}(x)$ ))  $\Rightarrow$  Q a"
3 proof -
4   have q: "a N  $\Rightarrow$  Q 0  $\Rightarrow$  ( $\wedge x. x \mathbf{N} \Rightarrow$  ( $\wedge y. y \mathbf{N} \Rightarrow y \leq x = 1 \Rightarrow Q y$ )  $\Rightarrow$  ( $Q \mathbf{S}(x)$ ))  $\Rightarrow$ 
5      $\forall x. (x \leq a = 1) \rightarrow Q x$ "
6   apply (rule ind[where a="a"], assumption)
7   apply (rule forallI implI)+
8   apply (rule eqBool, rule leq_terminates, assumption)
9   apply (rule nat0, rule natS, rule nat0)
10  proof -
11    fix x
12    show "x N  $\Rightarrow$  x  $\leq$  0 = 1  $\Rightarrow$  Q 0  $\Rightarrow$  Q x"
13      apply (rule eqSubst[where a="0" and b="x"], rule eqSym)
14      apply (rule leq_0, assumption+)
15      done
16    show "a N  $\Rightarrow$  x N  $\Rightarrow$ 
17      Q 0  $\Rightarrow$ 
18      ( $\wedge x. x \mathbf{N} \Rightarrow$  ( $\wedge y. y \mathbf{N} \Rightarrow y \leq x = 1 \Rightarrow Q y$ )  $\Rightarrow$  Q ( $\mathbf{S} x$ ))  $\Rightarrow$ 
19       $\forall xa. xa \leq x = 1 \rightarrow Q xa \Rightarrow$ 
20       $\forall xa. xa \leq (\mathbf{S} x) = 1 \rightarrow Q xa$ "
21    apply (rule forallI implI)+
22    apply (rule eqBool, rule leq_terminates, assumption)
23    apply (rule natS, assumption)
24    apply (rule natS, rule nat0)
25    proof -

```

```

26      fix xa
27      assume xa_nat: "xa N"
28      assume hyp: " $\forall x'. x' \leq x = 1 \rightarrow Q x'$ "
29      assume step: " $(\wedge x. x N \Rightarrow (\wedge y. y N \Rightarrow y \leq x = 1 \Rightarrow Q y) \Rightarrow Q (S x))$ "
30      assume xa_leq_sx: " $xa \leq S x = 1$ "
31      have H: " $xa \leq x = 1 \rightarrow Q xa$ "
32      by (rule forallE[where a="xa"], rule hyp, rule xa_nat)
33      show "x N  $\Rightarrow$  Q xa"
34      apply (rule disjE1[where P=" $xa \leq x = 1$ " and Q=" $\neg xa \leq x = 1$ "])
35      apply (fold GD.bJudg_def)
36      apply (rule eqBool)
37      apply (rule leq_terminates)
38      apply (rule xa_nat, assumption)
39      apply (rule natS, rule nat0)
40      apply (rule implE[where a=" $xa \leq x = 1$ "])
41      apply (rule H)
42      apply (assumption)
43      proof -
44      assume xa_not_leq_x: " $\neg xa \leq x = 1$ "
45      have xa_eq_sx: " $x N \Rightarrow xa = S x$ "
46      apply (rule leq_suc_not_leq_implies_eq)
47      apply (rule xa_nat, assumption)
48      apply (rule xa_not_leq_x)
49      apply (rule xa_leq_sx)
50      done
51      have q_sx: " $x N \Rightarrow Q S(x)$ "
52      apply (rule step)
53      apply (assumption)
54      apply (rule implE)
55      apply (rule forallE)
56      apply (rule hyp, assumption+)
57      done
58      show "x N  $\Rightarrow$  Q xa"
59      apply (rule eqSubst[where a="S x" and b="xa"])
60      apply (rule eqSym)
61      apply (rule xa_eq_sx, assumption)
62      apply (rule q_sx, assumption)
63      done
64      qed
65      qed
66      qed
67      assume step: " $(\wedge x. x N \Rightarrow (\wedge y. y N \Rightarrow y \leq x = 1 \Rightarrow Q y) \Rightarrow (Q S(x)))$ "
68      show "a N  $\Rightarrow$  Q 0  $\Rightarrow$  Q a"
69      apply (rule implE[where a=" $a \leq a = 1$ "])
70      apply (rule forallE[where Q=" $\lambda x. (x \leq a = 1) \rightarrow Q x$ "])

```

```

71      apply (rule q)
72      apply (assumption+)
73      apply (rule step)
74      apply (assumption+)
75      apply (rule leq_refl)
76      apply (assumption)
77      done
78 qed

```

■

It is necessary to use the object level connectives (i.e.  $\forall$  instead of  $\bigwedge$  and  $\rightarrow$  instead of  $\Rightarrow$ ) in the stronger proposition induction is performed over, since the induction axiom only works over expressions of type `o` (i.e. the object-level truth value type) and not of type `prop` (which the meta-level connectives  $\Rightarrow$  and  $\bigwedge$  are defined on). Thus, applying the GA induction axiom on the corresponding meta-level proposition  $\bigwedge x. (x \leq a = 1) \Rightarrow K x$  would not work. It is simply a type error.

The idea of this proof is again very straightforward, but spelling it out using the axioms is lengthy and challenging.

Using the strong induction lemma, termination of the division function defined earlier can be proven.

**Theorem 9.** *Termination of div*

$$\frac{\Gamma \vdash x \text{ N} \quad \Gamma \vdash y \text{ N}}{\Gamma \vdash \text{div } x \ y \text{ N}}$$

**Proof.**

By strong induction over the second argument.

```

1  lemma div_terminates:
2    shows "x N  $\Rightarrow$  y N  $\Rightarrow$  div x (S y) N"
3  apply (rule strong_induction[where a="x"], assumption)
4  apply (rule defE[OF div_def], simp)
5  proof -
6    fix x
7    assume hyp: " $\wedge y. ya \text{ N} \Rightarrow ya \leq x = 1 \Rightarrow (\text{div } ya \ (S \ y) \text{ N})"$ 
8    show "x N  $\Rightarrow$  y N  $\Rightarrow$  div (S x) (S y) N"
9      apply (rule defE[OF div_def])
10     apply (rule condT, simp)
11     apply (rule hyp, simp+)
12     done
13 qed

```

Isabelle

■



This proof is much shorter than the previous ones despite significantly higher complexity. The magic lies in the `simp` method, which invokes the simplifier. This theorem was proven much later than the previous termination proofs and using a lot of automation. The simplifier applies numerous previously proven lemmas here, for example to automatically solve the base case of  $\text{div } 0 \text{ S}(y) \text{ N}$ .

Section 4 goes into how automation is introduced into an axiomatized logic in *Pure*.

The authors of *GA* place a lot of emphasis on primitive recursion as a ‘benchmark’ for the expressivity of *GA*. Namely, they proved that all primitive recursive functions can be expressed and proven terminating in *GA*. While such a proof is out of reach for this formalization, it is more fitting for this formalization to show that *GA* can actually go beyond that. The Ackermann function is famously not primitive recursive [5]. With the tooling from Section 4, a termination proof of the Ackermann function is surprisingly simple to spell out in *GA*, using the standard approach of nested induction.

Consider the following standard definition of the Ackermann function in *GA*.

```
1 axiomatization
2   ack :: "num ⇒ num ⇒ num"
3 where
4   ack_def: "ack x y := if x = 0 then y + 1
5               else if y = 0 then ack (P x) 1
6               else ack (P x) (ack x (P y))"
```

Isabelle

**Theorem 10.** *Termination of ack*

$$\frac{\Gamma \vdash x \text{ N} \quad \Gamma \vdash y \text{ N}}{\Gamma \vdash \text{ack } x \ y \text{ N}}$$

**Proof.**

By nested induction using a helper lemma.

The outer induction ranges over the second argument and proves the stronger statement  $\forall n. \text{ack } m \ n \text{ N}$ .

```
1 lemma [simp]: "n N ⇒ ack 0 n = n + 1"
2 by (unfold_def ack_def, simp)
3
4 lemma "n N ⇒ m N ⇒ ack m n N"
5 apply (rule forallE[where a="n"])
6 apply (induct m)
7 apply (rule forallI, simp)
8 apply (rule forallI)
9 proof -
10   fix x z
11   show "x N ⇒ ∀y. ack x y N ⇒ z N ⇒ ack (S x) z N"
12     apply (induct z)
```

Isabelle

```
13    apply (unfold_def ack_def)
14    apply (subst rule: condI2, simp)
15    apply (subst rule: condI1, simp+)
16    apply (rule forallE[where a="1"], simp)
17    apply (rule forallE[where a="1"], simp)
18    apply (subst rule: condI1, simp+)
19    apply (rule forallE[where a="1"], simp)
20    apply (rule forallE[where a="1"], simp)
21    apply (unfold_def ack_def)
22    apply (subst rule: condI2, simp+)
23    apply (subst rule: condI2, simp+)
24    apply (rule forallE, simp)+
25    apply (rule forallI, simp+)
26    apply (rule forallE, simp)
27    done
28 qed
```

■

# Tooling for Isabelle/GA

4

Having implemented *GA* in *Pure* effectively obtained an interactive theorem prover we term *Isabelle/GA*, based on the axioms of *GA*. In its current state however, *Isabelle/GA* is not a very useful theorem prover. There is no proof automation, no term rewriting, and no easy way to formalize higher level mathematics. Users can only reason about natural numbers and only using axioms or previously proven lemmas, leading to highly verbose and cumbersome proofs, as seen in Section 3.

This chapter aims for making *Isabelle/GA* more usable as a proof assistant and, towards that end, introduces various methods for simpler and cleaner reasoning, a simple auto-solver, and most importantly, compatibility with the powerful simplifier built into *Pure*. The goal is to mostly hide the axiomatic system of *GA* behind abstract proof methods found in existing theorem provers and allow proofs to focus on their main idea, rather than being about mapping to specific axioms and discharging *habeas quid* premises.

## 4.1 (Un)folding (Recursive) Definitions

The first methods we implement are the `unfold_def` and `fold_def` methods, which take the name of a definition (`:=`) and (un)fold it once in the current goal state.

Example usage:

```
1 apply (unfold_def mult_def)
2 apply (fold_def mult_def)
```

Isabelle

This corresponds exactly to:

```
1 apply (rule defE[OF mult_def])
2 apply (rule defI[OF mult_def])
```

Isabelle

The method names are intentionally similar to the existing `unfold` and `fold`, which (un)fold an (non-recursive) Isabelle definition.

The implementation in SML is as follows:

```
1 structure Unfold_Def =
2 struct
3   fun fold_def_method thm_name ctxt =
4     SIMPLE_METHOD' (fn i =>
5       let
6         val defI_thm = Proof_Context.get_thm ctxt "defI"
7         val target_thm = Proof_Context.get_thm ctxt thm_name
8       in
9         CHANGED (resolve_tac ctxt [defI_thm OF [target_thm]] i)
10      end)
11
12   fun unfold_def_method thm_name ctxt =
```

SML

```

13   SIMPLE_METHOD' (fn i =>
14     let
15       val defE_thm = Proof_Context.get_thm ctxt "defE"
16       val target_thm = Proof_Context.get_thm ctxt thm_name
17     in
18       CHANGED (resolve_tac ctxt [defE_thm OF [target_thm]] i)
19     end)
20 end
21
22 val _ =
23   Theory.setup
24     (Method.setup @{binding unfold_def}
25       (Scan.lift Args.name >> Unfold_Def.unfold_def_method)
26       "Unfold a definition using defE"
27     )
28
29 val _ =
30   Theory.setup
31     (Method.setup @{binding fold_def}
32       (Scan.lift Args.name >> Unfold_Def.fold_def_method)
33       "Fold a definition using defI"
34     )

```

The Isabelle/ML infrastructure is well-equipped to handle such method definitions and multiple components of the provided infrastructure are visible in the snippet:

1. **Parsing:** `Args.name` parses an identifier and the `>>` combinator passes the parsed argument on to the defined method.
2. **Tactic Combinators:** `resolve_tac` simply applies the given list of theorems (in this case a singleton list) to subgoal `i` of the given context. `CHANGED` takes a tactic and succeeds if and only if its argument tactic changed the goal state. That is, if there was no definition to fold/unfold, the method fails, even if the theorem application itself succeeds.
3. **Method registration:** `Method.setup` sets up the defined method at the given binding. `SIMPLE_METHOD'` converts a value of type `int  $\Rightarrow$  tactic` (usually a function applying a tactic to the `i`'th subgoal, where `i` is its argument) to a method.

## 4.2 Configuring the Simplifier

*Pure* already contains a simplifier. The rewrite rules it uses are theorems of the shape:

```
1 complicated_expression  $\equiv$  simple_expression
```

Isabelle

That is, the *Pure* simplifier works on meta-level equations. When applied, the simplifier tries to match the left-hand side of any rewrite equation in any subexpression and rewrites it to the right-hand side of the equation.

The simplifier uses theorems tagged with `[simp]` as its rewrite rules. Its exact inner workings are not of interest here, but it is highly sophisticated and sublinear in the number of rewrite rules.

The first problem with using the simplifier in *GA* is that the axioms allow deriving object-level equality ( $=$ ), but not meta-level equality ( $\equiv$ ). Either, *GA* needs its own simplifier, or it needs to seek ‘compatibility’ of object equality with meta equality.

The simple solution is to add an axiom to convert object equality to meta equality:

### Equality Reflection

$$\frac{\Gamma \vdash a = b}{\Gamma \vdash a \equiv b}$$

This proposition is not provable from either the *Pure* axioms or the *GA* axioms, which is why it needs to be axiomatized. Using it for rewrites however is completely safe, as any such rewrite can be achieved using the existing `eqSubst` axiom.

If a rewrite theorem (that is, a theorem tagged with `[simp]`) has premises, the simplifier only rewrites if it can discharge all its premises.

The following SML structure configures the simplifier for *GA*. It implements three functionalities:

1. Converts object equality theorems to meta equality theorems on the fly.
2. Configures a solver tactic.
3. Sets a subgoal.

```

1  structure GD_Simp =
2  struct
3      fun reflect_eq th: thm = th RS @{thm eq_reflection};
4
5      fun match_object_rule th trm =
6          case trm of
7              Const (@{const_name "GD.eq"}, _) $ _ $ _ =>
8                  (case try reflect_eq th of SOME th' => [th'] | NONE => [])
9              | _ => []
10
11     fun mksimps _ th =
12         case Thm.concl_of th of
13             Const (@{const_name "Pure.eq"}, _) $ _ $ _ => [th]
14             | Const (@{const_name "GD.Trueprop"}, _) $ x => match_object_rule th x
15             | _ => []
16
17     fun step_tac ctxt i =
18         let
19             val close = reflexive_thm :: Simplifier.prem_of ctxt
20             val tac =

```

SML

```
21     assume_tac ctxt
22     ORELSE' resolve_tac ctxt close
23     ORELSE' GDAuto.gd_auto_tac ctxt
24   in
25     REPEAT_DETERM (CHANGED (tac i))
26   end
27
28   val gd_solver =
29     Raw_Simplifier.mk_solver "GD_solver" step_tac
30 end;
31
32 let
33   fun set_solver ctxt =
34     Raw_Simplifier.setSolver (ctxt, GD_Simp.gd_solver)
35   fun set_ssolver ctxt =
36     Raw_Simplifier.setSSolver (ctxt, GD_Simp.gd_solver)
37   fun configure ctxt =
38     ctxt
39     |> Simplifier.set_mkimps GD_Simp.mkimps
40     |> Simplifier.set_subgoaler asm_full_simp_tac
41     |> set_solver
42     |> set_ssolver
43 in
44   Theory.setup (Simplifier.map_theory_simpset configure)
45 end;
```

# Encoding Inductive Datatypes in GD

5

With Isabelle/GD now being a slightly more convenient proof assistant, the next goal is to make it easier to extend the domain of discourse beyond just natural numbers. Modern proof assistants, like Isabelle/HOL, contain fancy definitional mechanisms that allow for easy definition of things like inductive datatypes, recursive predicates, infinitary sets, and so on.

These definitional packages are effectively *theory compilers*, as they take a simple high-level definition, like an inductive datatype declaration, and map it to definitions, axioms, and automatically proven lemmas, encoding the high-level definition in lower level existing primitives.

The goal of this chapter is to take steps towards such a definitional mechanism for inductive datatypes in Isabelle/GD and encode them into the existing natural number theory. That is, any inductive datatype should be definable and conveniently usable without adding any axioms.

The roadmap towards this lofty goal is as follows:

- Formalize enough basic number theory to be able to define cantor pairings and some basic properties about them.
- Manually encode an inductive datatype into the natural numbers using the cantor pairing infrastructure from the first step. Define a type membership predicate, define the constructors as cantor pairings of their arguments and prove the necessary lemmas (such as all constructors being disjoint, the type membership predicate returning true for all values of the constructors, induction over the datatype, and so on).
- Plan out a semantic type system embedded within the single syntactic type of `num` in *Pure*.
- Write a definitional package that parses an inductive datatype declaration and compiles it into the necessary definitions, lemmas, and accompanying proofs.

## 5.1 Next

- cantor pairing formalization; cantor pairing and projection functions recursive formalizations and mutual termination proofs; key theorems

Before:

```

1 lemma cons_is_list:
2   assumes n_nat: "n N"
3   assumes xs_list: "is_list xs"
4   shows "is_list (Cons n xs)"
5   apply (rule eqSubst[where a="True"])
6   apply (unfold_def is_list_def)
7   apply (rule eqSym)
8   apply (rule condI2BEq)
9   apply (gd_auto)
10  apply (rule n_nat)
11  apply (rule list_nat)

```

Isabelle

```
12 apply (rule xs_list)
13 apply (gd_auto)
14 proof -
15   show "(if Cons n xs = Cons n xs ∧ is_list xs ∧ (n N) then True else False)
    = True"
16   apply (rule condI1B)
17   apply (rule conjI)+
18   apply (fold isNat_def)
19   apply (rule cons_nat)
20   apply (rule n_nat)
21   apply (rule list_nat)
22   apply (rule xs_list)
23   apply (rule xs_list)
24   apply (rule n_nat)
25   apply (rule true_bool)
26   done
27   show "True" by (rule true)
28 qed
```

After:

```
1 lemma cons_is_list [gd_auto]:
2   shows "n N ⇒ xs N ⇒ is_list xs ⇒ is_list (Cons n xs)"
3 by (unfold_def is_list_def, simp)
```

Isabelle



# A References

- [1] B. Ford, “Reasoning Around Paradox with Grounded Deduction.” [Online]. Available: <https://arxiv.org/abs/2409.08243>
- [2] L. C. Paulson, “Isabelle: The Next 700 Theorem Provers.” [Online]. Available: <https://arxiv.org/abs/cs/9301106>
- [3] L. C. Paulson, “The foundation of a generic theorem prover,” *J. Autom. Reason.*, vol. 5, no. 3, pp. 363–397, 1989, doi: 10.1007/BF00248324.
- [4] L. Paulson, T. Nipkow, and M. Wenzel, “The Isabelle Reference Manual,” 1998.
- [5] R. Péter, *Recursive functions in computer theory*. Ellis Horwood, 1981.